

# Príklad práce s tabukou (akcie PG\_ ...)

V příklade sa predpokladá existencia objektov [SD.DbStruct](#), [TB.Table](#). alej musí by spustený príslušný proces [D2000 DBManager](#) a v operanom systéme je v ODBC korektné konfigurovaná databáza s príslušným [DSN](#) menom.

```
INT _handle
INT _retCode
INT _pageSize
INT _rowCount
INT _numPages
INT _pgNr
TEXT _where
TEXT _orderBy
RECORD (SD.DbStruct) _dbPageArr
ALIAS (SD.DbStruct) _targetArr

SET _targetArr AS SV.Struktura
_pageSize := _targetArr\DIM

; otvorenie tabuky
PG_CONNECT TB.Table, _DB_MODIFY, _handle, _pageSize, _where, _orderBy, _rowCount, _retCode

IF _retCode # _ERR_NOERROR THEN ; nepodarilo sa spoji s tabukou
END
ENDIF

; ak nie je vo výbere ni => koním
IF _rowCount = 0 THEN
END
ENDIF

; vypoítam poet stránok
_numPages := _rowCount / _pageSize
IF _numPages*_pageSize < _rowCount THEN
_numPages := _numPages + 1
ENDIF

; prejdem postupne všetky stránky
_pgNr := 1 ; aktuálna stránka

DO_LOOP
PG_READ _handle, _pgNr, _dbPageArr, _retCode
GOSUB Copy_dbPageArr ; skopírujem stránku do _targetArr
_pgNr := _pgNr + 1
EXIT_LOOP _pgNr > _numPages
END_LOOP

PG_DISCONNECT _handle ; zatvorím tabuku
END

, *****
; Kopírovanie obsahu lokálnej premennej _dbPageArr do _targetArr
, *****
Copy_dbPageArr:
INT _itemsToCopy
INT _i

; zistím, koko prvkov budem kopírova
IF _targetArr\DIM < _dbPageArr\DIM THEN
_itemsToCopy := _targetArr\DIM
ELSE
_itemsToCopy := _dbPageArr\DIM
ENDIF

_i := 1
DO_LOOP
EXIT_LOOP _i > _itemsToCopy

SET _targetArr[_i] WITH _dbPageArr[_i]
_i := _i + 1
END_LOOP

WAIT ; pokám na uskutenenie posledného priradenia do _targetArr
RETURN
```

**Poznámka:**

Pri ukonení skriptu akciou [END](#) (alebo iným ubovoným spôsobom), sú automaticky zatvorené všetky spojenia s tabukou.

**Súvisiace stránky:**

[Akcie v skriptoch](#)